
News  

Spike2 and Signal Training Days 
This years U.K. Spike2 and Signal training courses are fast approaching and will take 
place on Thursday 30th April and Friday 1st of May in Cambridge. The Spike2 thread for 
both days is now fully booked, but there are still limited places available on the Signal 
and Workshop threads. If you or any of your colleagues would like to attend, please 
follow the links to view the training day program and complete the registration form. 
Please give a short description of the type of research that you do and analysis that you 
are interested in. This will allow us to tailor sessions for specific requirements. 

Opening and saving embedded files 
The scripts, sequence files and sampling configurations included in the email newsletter 
are embedded files, which can be opened or saved to disk by right-clicking on the 
following style of icons which are positioned next to the section title: 

 
 

 

Embedded script icon showing right-click context menu 

Please let us know if you have any problems opening or saving these embedded files. 

Meetings and events 
Experimental Biology 2009 
Ernest N. Morial Convention Center  
New Orleans, LA, USA 
April 18th – 22nd 2009 

British Neuroscience Association 20th National Meeting 
Adelphi Hotel  
Liverpool, UK 
April 19th – 22nd 2009 

 

 

 

International Workshop and Conference on Human 
Reflexes: Wiring and Firing of Motorneurons 
Ege University  
Izmir, Turkey 
May 11th – 15th 2009 

3rd Annual Canadian Neuroscience Meeting 
Westin Bayshore 
Vancouver, Canada 
May 24th – 27th 2009 

 

 

 

The 6th Congress of the International Society for Autonomic 
Neuroscience 
ISAN09 is a joint meeting to be held with the European 
Federation of Autonomic Societies (EFAS) in Sydney, from 
September 1st-4th, 2009. Approx. 400 international and national 
delegates are expected, having clinical and/or basic scientific 
research interests in autonomic neuroscience. 

Plenary speakers: Darwin Berg (UCSD), Hugo Critchley (University of Sussex), Max Hilz 
(University of Erlangen-Nuremberg) and Elspeth McLachlan (POWRI) 

Further details of the meeting can be found at http://www.iceaustralia.com/isan2009/ 

Proudly supported by Cambridge Electronic Design and the Foundation for High Blood 
Pressure Research, Australia. 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 6.11 Signal version 4.05 

Spike2 version 5.21 Signal version 3.11 

Spike2 demo Signal demo 

 

Spike2   
Q. I am currently using PCA clustering to sort spikes off-line. One very common 

result of this is that I quickly find clusters of spikes that are well separated and 
assign them a code. I then use the Apply changes function from the cluster 
window to classify these spikes in the data file. I am then left with another cluster 
in which I can see different units that are not as easily separated. What I would 
like to do is re-run the PCA only on this second cluster, without the spikes that I 
have already sorted contributing to the analysis. Is there an easy way to do this?   

A. The marker filter function in the Spike2 analysis menu can be used to show or 
hide classified and non-classified spike shapes in the data file. Any analysis 
functions performed on a WaveMark channel will then only use the visible spikes. 
In this example, after applying the changes back to the data file, simply use the 
marker filter to hide the easily separated spikes in the data file. You can then use 
the Reanalyse function from the File menu in the cluster window to re-run the 
PCA analysis using only the visible spikes from your original WaveMark channel. 

Reanalyse function in the cluster dialog 

 

  

 

Q. How do I use multiple states to control Magstim settings?   

A. Multiple states control of supported Magstim devices is achieved using the 
auxiliary states mode in Signal. The option of including auxiliary states device 
support is given during the Signal installation. 

 When the Multiple states checkbox is enabled in the General tab of the sampling 
configuration and Dynamic outputs mode is selected, the Magstim button in the 
States tab allows the user to configure settings for Magstim 200, BiStim or Rapid 
devices. Depending on the type of Magstim selected, each enabled state 
includes settings for Power %, no. of pulses and frequency of stimulation (Hz). 
During sampling, these settings are applied to the Magstim device for the current 
state. Signal can switch between states in numeric or random order during an 
experiment, or run multiple user-defined protocols for state sequencing. 

  
Configuration settings for a Magstim Rapid device 

 Further information on Magstim control using multiple states can be found in the 
on-line help. 

CED
You can copy clustering data as text from the Edit menu of the cluster window

CED
Magstim settings are saved as frame variables in the data file. You can access frame variable values using the script language



Scripts: Spike2  
Q. I am using Spike2 to record ECG and would like to know if there is any way to 

output a tone when an ECG complex occurs, much like a standard ECG monitor? 

A. The attached script, ECGTone.s2s, plays a tone pip from DAC 0 of the 1401 
when the R-wave of an ECG complex is detected during sampling. Sampling is 
controlled via the script toolbar and the user is asked to set the waveform 
channel to monitor and a hysteresis value for peak detection. A horizontal cursor 
is then used to set a threshold for the detection of the R-wave. Each time the 
threshold is crossed Spike2 monitors the level of the incoming data and, when a 
peak is found, waits for the data to fall by at least the level set as the hysteresis 
value before playing the tone which flags the R wave detection. 

 This script requires Spike2 version 6 and writes a sequence file into a default 
folder of C:\Spike6\Sequence.  

Scripts: Signal  
Q. I have two states set-up in my sampling configuration to output single or double 

stimulation pulses during experiments. As I use peri-trigger sampling mode, it 
would be very useful if I could control which state to trigger from the digital inputs 
at the start of a sampling sweep. Is this possible with a script? 

A. State branching in response to digital inputs can be achieved by using the output 
sequencer in Signal. The attached sequence file and sampling configuration is an 
example of controlling state branching and pulse outputs in response to digital 
inputs at the beginning of a sampling sweep. The sequence file monitors the 
digital inputs during sampling and outputs a single pulse in response to bit 0 
being set high, or a double stimulation pulse in response to bit 1 being set high. 
The sequence also sets the relevant frame state (1 for single stimulation and 2 
for double). 

 To run the sequence, save the attached sequence file (DigBranch.pls) and 
sampling configuration (DigBranch.SGC). Load the sampling configuration 
from the File menu in Signal and browse to the saved sequence file from the 
Outputs tab. The sequence could easily be expanded to control additional states 
and outputs if required. 

 
Sampling configuration with linked sequence file 

If you have any problems 
opening the embedded 
scripts in this newsletter 

'ECGTone.s2s
'This is a script to output a tone upon detection of a peak or trough 
'on a waveform input channel. 
'
'Note this script is UNDER DEVELOPMENT BY CED and is included for
'illustration only. We make no warranties as to its suitability
'or otherwise. It has not been thoroughly tested.
'You should use it 'as is'

''CED 18/04/09

'The script uses the current sampling configuration so a suitable waveform channel must be set up.

'NOTE: This script writes an output sequence into a default folder of c:\Spike6\sequence into which 
'variables are passed. If this folder does not exist the script will not work. The script requires Spike2 version 6 to run.


var data%;								'Handle of new data file
var sTime;								'The last time we looked at the idle routine
var curtrigLev;						'Current trigger level as checked by idle routine		
var oldTrigLev;						'Previous trigger level
var trigChan%;							'Memory channel used for testing purposes
var inputChan%;						'Waveform channel monitored for threshold crossings
var setHyst;							'Delay time in ms
var hystVar;
var delayTicks%;						'Delay in sequencer steps to pass as variable
var trigVar;							'Level of threshold in ADC units
var threshType%;
var ToneOut%;
var output%;
var Length:=1;

ToolbarVisible(1);					'Make toolbar visible always
New%();									'Set up new sampling window
DoToolbar();							'Do the toolbar

func New%()								'New sampling window
var ok%;
View(LogHandle());						'Make log view the current view
EditSelectAll();						'Select all text in log view
EditClear();							'Delete it
Window(0,80,100,100);					'Display it at the bottom of the screen
WindowVisible(1);						'Make it visbible

if data%>0 then							'If there is already a data view open then
  if ViewKind(data%)>-1 then
  	 View(data%);							'Close it
  	 FileClose();
  endif;
endif;


DlgCreate("Output parameters");				'Dialog to set channel and delay
DlgInteger(1,"Input channel",1,32);
DlgReal(2,"Hysteresis",0,1000);
DlgList(3,"Threshold","Peak|Trough");
ok% := DlgShow(inputChan%,setHyst,threshType%);
if not ok% then
	Message("Pressed cancel - Script closing");
	Halt;
endif;

output%:=FileNew(7,1,1,1,Length,32); ''''New file holding output data
Window(25,0,75,50);

ToneOut%:= VirtualChan(0, "WEnv(0.1, 0.2, 0.1, 0)*WSin(1000, 0)", 0,0.0001,0);  'Edit virtual channel
ChanShow(ToneOut%);
PlayWaveAdd("T", "Tone",0,0.0, MaxTime(),ToneOut%,1); 'Add the channel from the output file to on-line output
FileClose(0,-1);

WriteSequence(threshType%);									'Write output sequence to use

data%:=FileNew(0,2);					'Open a new data file for sampling
HCursorDelete(-1);
WindowVisible(1);
if data%<0 then Message("Unable to open new data file");Halt() endif;
'DrawMode(-1,2);							'Set event draw mode to lines
Window(0,0,100,100);						'Make data window in top bit of screen
'XRange(0,10);
HCursorNew(inputChan%,oldTrigLev);				'Set cursor on appropriate channel	
Interact("Position cursor at initial threshold level",1023);
oldTrigLev:=HCursor(1);					'Get threshold level
trigVar:=(HCursor(1)-ChanOffset(inputChan%))/(5*ChanScale(inputChan%))*32768;		'Convert to ADC units
SampleSeqVar(1,trigVar);				'Set output sequencer variable
hystVar := (setHyst/ChanScale(inputChan%)) * (65536/10);
SampleSeqVar(2,hystVar);
hystVar := setHyst/ChanScale(inputChan%);
trigChan%:=MemChan(3);	
'FrontView(LogHandle());					'Bring the Log view to the front
FrontView(data%);						'Bring the data view to the front
ToolbarEnable(3,0);						'Disable "Sample stop" button
ToolbarEnable(2,1);						'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(1,"Quit",Quit%);				'Set up toolbar buttons
ToolbarSet(2,"Sample start", Start%);
ToolbarSet(3,"Sample stop", Stop%);
ToolbarSet(4,"New file", New%);
'ToolbarSet(6,"Change delay",ChangeDelay%);
ToolbarEnable(3,0);						'Disable "Sample stop" button
ToolbarEnable(6,0);
View(data%);
Toolbar("Press SAMPLE START to commence sampling", 1023);					'Wait here until quit is pressed
end;

func Quit%()							'If "Quit" is pressed
SampleStop();							'Stop sampling
return 0;								'leave toolbar
end;

func Start%()							'If "Start" is pressed
ToolbarSet(0,"",Idle%);					'Call Idle%() whenever there is free time
SampleStart();							'Start sampling
ToolbarEnable(4,0);						'Disable "New file" button
ToolbarEnable(3,1);						'Enable "Sample stop" button
ToolbarEnable(2,0);						'Disable "Sample start" button
ToolbarEnable(1,0);						'Disable "Quit" button
ToolbarEnable(6,1);
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;								'Stay with toolbar
end;

func Stop%()							'If "Stop" is pressed
ToolbarClear(0);
SampleStop();							'Stop sampling
ToolbarEnable(4,1);						'Enable "New file" button
ToolbarEnable(3,0);						'Disable "Sample stop" button
ToolbarEnable(1,1);						'Enable "Quit" button
ToolbarEnable(6,0);
ToolbarText("Press FILE NEW to capture more data");
return 1;								'Stay in toolbar
end;


'Func ChangeDelay%()													'Function to change delay time
'setDelay:=Input("Delay time (ms)",setDelay,0.1,1000);
'delayTicks%:=(setDelay*100-4);
'SampleSeqVar(4,delayTicks%);
'return 1;
'end;

func Idle%()							'The Idle routine is called when PC has time
var eTime;
View(data%);
eTime:=Maxtime();

'PLACE CODE IN HERE TO ANALYSE THE DATA FILE BETWEEN sTime and eTime
curTrigLev:=HCursor(1);								'Check if threshold level has changed
if curTrigLev <> oldTrigLev then					'If so, Get new level in ADC units
	trigVar:=(curTrigLev-ChanOffset(inputChan%))/(5*ChanScale(inputChan%))*32768;
	SampleSeqVar(1,trigVar);						'Update sequencer variables		
	oldTrigLev:=curTrigLev;							'Reset previous cursor level to current level
endif;
MemImport(trigChan%,1,sTime,eTime,1,0,0.3);					'For testing purposes

sTime:=eTime;
return 1;								'Stay in toolbar
end;


'This function writes an output sequnce for use with the script. 
'This is done as there is no other way to set the waveform channel to monitor directly
Func WriteSequence(type%)
var seq%;
var seqSet%;
seq%:=FileOpen("c:\\Spike6\\Sequence\\ECGTone.pls",8,1);
'Message(seq%);
View(seq%);
if type% = 0 then

Print("SET      0.0100 1 0\n");
Print("VAR    V1,thresh=VDAC16(0.75) ;Threshold level\n");
Print("VAR    V2,hyst			;Hysteresis amplitude\n");
Print("VAR    V3,counter=1000 ;Counter to set timeout\n");
Print("VAR    V4,current=0    ;Variable for current level on channel\n");
Print("VAR    V5,max=0        ;Max level achieved\n");
Print("VAR    V6,level\n");
Print("VAR    V7,ch\n\n");
Print("; The initial setup waits for us to be below the threshold\n");
Print("OVER:       CHAN   current,1       ;Get current level on channel\n");
Print("            BGT    current,thresh,over ;If above, keep looking\n");
Print("; Then we wait for us to be above the threshold\n");
Print("UNDER:      CHAN   current,1       ;Get current level\n");
Print("            BLT    current,thresh,under ;If under, keep looking\n");
Print(";Resets max and level variable to track when waveform is rising\n");
Print("SETMAX:     MOV    max,current     ;Set max to current level\n");
Print("            MOV    level,max       ;Set level to max - hysteresis\n");
Print("            SUB    level,hyst\n");
Print("            LDCNT1 counter         ;Load counter to loop through for set time\n");
Print("; The core loop. If below trigger level do the pulse, if above max then reset\n");
Print("; the max value and trigger level, otherwise count down the timeout\n");
Print("RISE:       CHAN   current,1       ;Get current level\n");
Print("            BLT    current,level,tone ;If lower than required level (max - hyst) output tone\n");
Print("            BGT    current,max,setmax ;If new maximum, reset the levels\n");
Print("            DBNZ1  rise\n");
Print("            JUMP   over\n");
Print("TONE:      WAVEGO   T\n");
'Print("              DELAY    1999\n"); 
'Print("            DIGOUT [00000000]\n");
Print("            Report\n");
Print("            JUMP   over\n");


else


Print("SET      0.0100 1 0\n");
Print("VAR    V1,thresh=VDAC16(0.75) ;Threshold level\n");
Print("VAR    V2,hyst			;Hysteresis amplitude\n");
Print("VAR    V3,counter=1000 ;Counter to set timeout\n");
Print("VAR    V4,current=0    ;Variable for current level on channel\n");
Print("VAR    V5,max=0        ;Max level achieved\n");
Print("VAR    V6,level\n");
Print("VAR    V7,ch\n\n");
Print("; The initial setup waits for us to be below the threshold\n");
Print("OVER:       CHAN   current,1       ;Get current level on channel\n");
Print("            BLT    current,thresh,over ;If below, keep looking\n");
Print("; Then we wait for us to be above the threshold\n");
Print("UNDER:      CHAN   current,1       ;Get current level\n");
Print("            BGT    current,thresh,under ;If over, keep looking\n");
Print(";Resets max and level variable to track when waveform is rising\n");
Print("SETMAX:     MOV    max,current     ;Set max to current level\n");
Print("            MOV    level,max       ;Set level to max + hysteresis\n");
Print("            ADD    level,hyst\n");
Print("            LDCNT1 counter         ;Load counter to loop through for set time\n");
Print("; The core loop. If below trigger level do the pulse, if above max then reset\n");
Print("; the max value and trigger level, otherwise count down the timeout\n");
Print("RISE:       CHAN   current,1       ;Get current level\n");
Print("            BGT    current,level,tone ;If greater than required level (max - hyst) output tone\n");
Print("            BLT    current,max,setmax ;If new minimum, reset the levels\n");
Print("            DBNZ1  rise\n");
Print("            JUMP   over\n");
    Print("TONE:      WAVEGO   T\n");
'    Print("              DELAY    1999\n"); 
'Print("            DIGOUT [00000000]\n");
Print("            JUMP   over\n");

endif;
'FileSaveAs("c:\\Spike4\\Sequence\\Pulse.pls");
FileSave();
FileClose();
seqSet%:=SampleSequencer("c:\\Spike6\\Sequence\\ECGTone.pls");
'Message(seqSet%);
return 1;
end;



CED
ECGTone.s2s

DigBranch.SGC



Did you know…? 
The virtual channels dialog in Spike2 holds a list of previously used expressions. 

 

 

 
Virtual channel dialog menu 

Recent questions   
Q. Is there any way to convert a Spike2 data file to an audio format for presentation 

purposes?  

A. The attached script, SMRtoWav.s2s, will take data from a Spike2 data file and 
save it to WAV audio file format.  

User group      
Try the new CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

'=============================================================================================================
'CAMBRIDGE ELECTRONIC DESIGN LIMITED, THE SCIENCE PARK, MILTON RD., CAMBRIDGE CB0 0FE, UK
'=============================================================================================================

'SmrToWav1.smr
'Copyright © Cambridge Electronic Design,  Mar 2003.

'Overview.
'
'This script converts waveform channels in .smr files created by CED Spike2 software to .wav format.
'It requires Spike 2 version 3 or higher.
'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'User Guide.

'Use the Run or Load and Run commands on the Spike 2 Script menu to start the script.  
'The main script toolbar has 8 buttons that can be clicked with the mouse or operated by a hotkey (underlined).

'Use the |Select View| button to choose a file to analyse from a list of already open .smr files
' or press |New File| to load a  data file from disk.
'Display the time range of interest using the usual Spike 2 controls, e.g. by scrolling, double-clicking the X-axis
'or clicking on the buttons in the bottom left hand corner of the time view. 
'When the area of interest is displayed, bracket the time range to convert with the cursors labelled 'start' and 'end'.
'If the cursors are not visible you can bring them into view with the |Fetch Cursors| button. 
'Alternatively, press the |Whole File| button to convert the entire file.

'Select the channels to be included in the .wav file by clicking on the channel numbers. Hold down Cntrl to select multiple channels.
'Finally, press the |Export as .wav| button. A dialog appears with a drop-down menu to select 8-bit or 16-bit integer
'format. Checkboxes indicate the currently selected waveform channels. Ensure that the required channels are selected
'(maximum 4), then press OK. Note that only channels with identical sampling rates can be converted.
'Press OK when ready. Error messages are shown if too many channels were selected, or if selected channels have different sampling rates.
'If all is well,a File Save dialog appears for entering the name and path of the output file. You must explicitly give the file
'a '.wav' filetype. Press Save to generate the output file.
 
'You can repeat this sequence with different time ranges and/or files as often as required, creating a new .wav file each time 
'Press |Quit| to close the script.

'The script has a limited .wav file playback facility.
'You can play back the most recently saved .wav file via the computer's sound system provided that the file is relatively small (<1.5MB)
'and has one  or two channels. To do this, click the PlayBack button.  An error message is displayed if the .wav file is not suitable.
'Larger files can be replayed with a media player.
'The |StopPlay| button interrupts playback of a wav file.

'****************************************************************************

	const BSZ% := 8000;                    'limit array to the size that can be written quickly
    const swquit%:=1, swselvw%:=4, swnewf%:=5;'button numbers
    const swfetch%:=7,swall%:=8,swstore%:=10;
    const swplay%:=12,swpstop%:=13,on%:=1,off%:=0;

	var   fh%, bh%;
	var   st, fin;
	var   gFloat%[20];                     		'global for floating window states
	var   views%[100];
	var outfname$;								'name of most recent .wav file
    var   playflg%:=0;                          'flag that  wav file can be played back with Sound()
    var   nbufs%:=0;                            'number of 8K buffers

'****************************************************************************

	ToolbarClear();
	View(App(3)).WindowVisible(2);							'iconise the script
	HideAll();
    if ViewKind(View(0)) = 0 then                           'if the current view is a cfs file use it as default 
       fh%:=View(0);
    WindowVisible(1);
    FetchCursors%();
    else                                                    'else find another open time view if available
        var list%[2],nc%;
        nc%:=ViewList(list%[],1);
        if nc% > 0 then
            fh%:=list%[1];
            View(fh%);
        WindowVisible(1);
         endif;
    endif;
    ToolbarSet(0,           "Idle",             Idle%);
    ToolbarSet(swquit%,     "&Quit",            Quit%);
    ToolbarSet(swselvw%,    "Select &View ",    Selview%);
    ToolbarSet(swnewf%,     "&New File",        OpenFile%);
    ToolbarSet(swfetch%,    "&Fetch Cursors",   FetchCursors%);
    ToolbarSet(swall%,      "&Whole File",     WholeFile%);
    ToolbarSet(swstore%,    "&Export as .Wav ", Storewav%);
	ToolbarSet(swplay%,     "&PlayLast ",       PlayWav%);
	ToolbarSet(swpstop%,    "&StopPlay ",       StopPlay%);
    ToolbarEnable(swplay%,off%);                                 'disable the playback buttons
    ToolbarEnable(swpstop%,off%);
    Toolbar(" Drag cursors to select time range to export", 191);
	halt;

'****************************************************************************

    func Idle%()
    var VL%,enable% := 0;
    var x%[2],n%;							                       	'temporary array to get size of chans%[]
	n%:=ChanList(x%[], 1+512);
    var chans%[n%+1];											'list of waveform/realwave channels
    VL%:= Viewlist(views%[], 1);
    
    if VL% < 2 then                                             'disable view select button if only one available
        ToolbarEnable(swselvw%,off%); 
    else 
        ToolbarEnable(swselvw%,on%); 
    endif; 
    
    if vl% > 0 then                                             'enable most buttons if at least one  .cfs document open
      enable% := 1; 
    else 
        ToolbarText("");                                        'clear toolbartext if no file
    endif; 
    ToolbarEnable(swfetch%,  enable%);
    ToolbarEnable(swall%,  enable%);
    
    if ((ViewKind(View(0)) = 0) AND (View(0).ChanList(chans%[], 1+512) = 0)) then  'no waveform/realwave channels 
       enable% := 0; 
    endif; 
    ToolbarEnable(swstore%, enable%);
    
    if ViewKind(View(0)) = 0  then                                  'Renumber and relabel cursors if they are 'crossed'
        if Cursor(1) > Cursor(2) then ' 
            CursorRenumber(); 
            DoCursors%(); 
        endif; 
    endif;
    return 1;
    end;

'****************************************************************************

func PlayWav%();
    if nbufs% < 200 then                                                'if the file is relatively small
	   Sound(outfname$,1);                                              ' play the wav file
    else
        Message("Sorry. Spike2 couldn't replay "+ outfname$ +"\n"+
        "Try using your media player.");
    endif;
        ToolbarEnable(swpstop%,on%);
	return 1;
	end;

'****************************************************************************

    func StopPlay%();
	Sound("S!",1);'stop previous sound output and play system exclamation sound.
    ToolbarEnable(swpstop%,off%);
	return 1;
	end;

'****************************************************************************

    func FetchCursors%()									'Places cursors to be in middle of view
    var vh%; 
    View(fh%);
    CursorSet(2);
    DoCursors%(); 
    return 1;
    end;

'****************************************************************************

	func WholeFile%()									   'Places cursors to be in the start and end of the view
	View(fh%).CursorSet(2, 0, View(fh%).MaxTime());
	return 1;
	end;

'****************************************************************************

    func SetView%()                                       'Set this view current
    view(fh%);
	Window(0,0,100,100);
  	Draw(0, MaxTime());
    CursorSet(2);
    DoCursors%(); 
    ToolbarText(" Drag cursors to select time range to export"); 
	FrontView(fh%);
	return 1;
    end;

'****************************************************************************
    func DoCursors%();										'label 2 cursors 
    var vh%; 
    vh%:=View(); 
    view(fh%); 
    CursorLabel(4,1,"start(%5.2ps )"); 
    CursorLabel(4,2,"end(%5.2ps )"); 
    CursorLabelPos(1,1); 
    CursorLabelPos(2,3); 
    View(vh%); 
    return 1; 
    end;
'****************************************************************************

	func SelView%()
	var i%, ok%, n%;
	
	for i% := 0 to 99 do
		views%[i%] := 0;
	next;
	if (Viewlist(views%[], 1) > 0) then                 	'Store viewhandles of all available time views
		var list$[views%[0]];
		for i%:=1 to views%[0] do
			view(views%[i%]);
			list$[i%-1] := windowtitle$();                	'Set list of titles
		next;
		dlgcreate("Select view");                        	'Define Menu ...
		dlglist(1,"Select view: ",list$[]);
		if (dlgshow(n%) = 0) then
			return 1;
		endif;
		fh%:=(views%[n%+1]);
		if (fh% > 0) then
			return SetView%();                             	'Activate view        
		endif;
	else 
		return OpenFile%();                              	'Open new file
	endif;
	return 1;
	end;

'****************************************************************************

	func OpenFile%()
	var vh%;
		vh% := FileOpen("",0,0,"Select a .smr file for conversion");
	if (vh% < 0) then
	   Sound("S*",1); 
	   return 1;
	else
	   fh% := vh%;
	   return SetView%();
	endif;
	end;

'****************************************************************************

	func Storewav%();
	var   ok%, bytes%, fmt% := 1;
	var   x%[1];							                 ' temporary array to get size of chans%[]
	var   sampInt;
	
	View(fh%);
	CursorRenumber();
	st  := Cursor(1);
	fin := Cursor(2);
	var chans%[ChanList(x%[],1+512)+1];						'array large enough to hold all waveform/realwave channels 
	
	ChanList(chans%[], 1+512);								'store list of wavefor/realwave ch
	ok% := DoConfig%(chans%[], fmt%);                        ' Put up the configuration dialog
	if ok% <> 0 then                                         'dump if user selects a channel
		bh% := FileOpen("", 9, 1, "Filename: Remember to include the extension '.wav'"); 
		outfname$:=Filename$();
		if bh% > 0 then
			ToolbarText("Saving .wav file");
			view(fh%);
			sampInt := BinSize(chans%[1]);
			docase
				case fmt% = 0 then bytes% := 1;
				case fmt% = 1 then bytes% := 2;
			endcase;
			SetupAndWriteHeader(bh%, sampInt, (fin-st)/sampInt, bytes%, chans%[0]);   'setup file header
			Dumpwave(bh%,  fh%, st, fin, chans%[0], bytes%, chans%[]);       '8/16-bit Integer 
			Sound("S!",1);
			View(bh%);
			FileClose();                                        				'close the binary file
            if playflg% > 0 then
                ToolbarEnable(swplay%,on%);                         			'enable the playback buttons
            endif;
            ToolbarText(" Drag cursors to select time range to export");		'change prompt
		else
			Sound("S*",1);
		endif ;
	endif;
	return 1;
	end

'****************************************************************************
	func DoConfig%(chans%[], &fmt%)
	var numchans%;
		numchans% := chans%[0];				        				' store number of channels
	var chansel%[numchans%]; 
	var ch%;
	var curchn% := 1;								                ' how many channels are selected?
	var ret%,									                    ' return value from DlgShow to be returned
		nocurs% := 1,								               	' should we not use cursors to find st and fin
		sep;										                ' which separator will we be using?
	var fmt$[4];
	var side% := 0, y%;
	var z; z := (numchans%+10.2)/2.0;
	                   ' z is the y position (in DBUs) of the next line after the channel list.
	                   ' z := (numchans%+2*row+1)/2.0; where row is the row the items to follow would be
	                   ' on if we had no channel list.
	
	                                                				' Start new dialog, specifying w + h
	fmt$[0] := "8-bit Integer"; fmt$[1] := "16-bit Integer"; fmt$[2] := "32-bit IEEE real"; fmt$[3] := "64-bit IEEE real";
	
	DlgCreate ("Export in .Wav format",0,0,57,((numchans%+1)/2)+8); 
	DlgLabel  (1,"Output Format",2,1.3);
	DlgList   (2,13,fmt$[],2,18,1.3);							    ' create listbox
	DlgLabel  (3,"Choose the channels to export",2,3); 
	
	' We now loop through all the channels and create a checkbox for each. We arrange
	' them into two columns to decrease the height of the dialog box.
	
	for y% := 1 to numchans% do
		ch% := chans%[y%];
	    chansel%[y%-1]:=ChanSelect(ch%);                           'check the box if channel was selected
		DlgCheck(y%+3,Print$("%3d %s %.2f KHz",ch%,ChanTitle$(ch%), 1.0/BinSize(ch%)/1000.0),25*side%+6,(y%+1)/2+3);
		side% := not side%;
	next;
	    if ArrSum(chansel%[]) =0 then                                   'if no channels selected set 1st by default
	        chansel%[0]:=1;
	    endif;
	var txt$ := "This script is not able to export files which contain gaps on one\n"
	   "data channel only. It can, however, process files with gaps on\n"
		   "all channels simultaneously correctly. All selected channels must\n"
	   "have the same sample rate";
	
	DlgLabel(numchans%+4, txt$, 3, z);
	ret% := DlgShow(z,fmt%,z, chansel%[], z);               		' ret% is 0 if user cancels
	if (ret% > 0) then
		for y% := 1 to numchans% do					           		' count them and eliminate those
			if chansel%[y%-1] then						            ' which aren't
				chans%[curchn%] := chans%[y%];
				curchn% += 1;
			endif;
		next;
		chans%[0] := curchn%-1;							           ' save number of selected channels
	    if chans%[0] <=2 then
	        playflg%:=1;                                           'flag that this file can be played back by the script
	    else
	        playflg%:=0;
	    endif; 
		if chans%[0] > 4 then 
			Message("Error|Can't export more than 4 channels.\n"+
			"Please go back and select 1 to 4 channels,"); 
			return 0; 
		endif;
	
		if chans%[0] = 0 then							           ' validate data; must have
			Message("Error|You must select a waveform channel to export!");
			return 0;									              ' at least one selected chan
		endif;
		if (chans%[0] > 1) then							           ' validate data; must have the same sampling rate
			for y% := 1 to chans%[0]-1 do					             ' count them and eliminate those
				if (View(fh%).BinSize(chans%[y%]) <> View(fh%).BinSize(chans%[y%+1])) then
		   			Message("Error|You must select channels with the same sampling rate!");
		   			return 0;									         ' at least one selected chan
				endif;
			next;
		endif;
		if st = fin then								               ' must have a nonzero period
			Message("Error|You must select a time region to export!"); 
			return 0;									               ' of time to dump
		endif;
	endif;
	return ret%;									                  ' return appropriate value
	end;


'****************************************************************************

	Proc SetupAndWriteHeader(bh%, sampInt, arrSiz%, bytes%, chans%);         ' Standard components of wav file header in integer format
	var   MyHead%[11];                     'array to store WAV format file header
	   Myhead%[0]  := 0x46464952;                               ' RIFF
	   MyHead%[1]  := 44 + (arrSiz%*bytes%*chans%);
	   Myhead%[2]  := 0x45564157;                               ' WAVE
	   Myhead%[3]  := 0x20746D66;                               ' fmt<sp>
	   Myhead%[4]  := 0x10;                                     ' formatsize=16,
	   Myhead%[5]  := 0x10000 * chans% + 0x01;                  ' upper 16 == chans, lower 16 == format tag == 1
	   MyHead%[6]  := 1.0 / sampInt;                            ' Fill in the gaps in the header array
	   MyHead%[7]  := MyHead%[6] * bytes% * chans%;             ' average bytes/second
	   Myhead%[8]  := (0x80000 * bytes%) + (bytes%* chans%);    ' upper 16 == bits/sample, lowest 16 == block allignment
	   Myhead%[9]  := 0x61746164;                               ' DATA
	   MyHead%[10] := arrSiz% * bytes% * chans%;
	
	   View(bh%).Bwrite(MyHead%[]);                             'write the header
	end;

'****************************************************************************

	proc DumpWave(bh%, fh%, sTime, eTime, nch%, bytes%, list%[])
	                                  
	var myArray%[BSZ%];
	var dat%[nch%][BSZ%];                                          'generate channels array
	var n%, ich%, pts%;
	
	repeat
		for iCh% := 0 to nCh%-1 do
			n% := ChanData(list%[ich%+1], myArray%[], sTime, eTime, stime);
			if n% > 0 then                                          'if we got data
				pts% := n%;
				if (bytes% = 1) then
					ArrDiv(myarray%[:n%], 256);
					ArrAdd(myarray%[:n%], 128);                      'using array arithmetic
				endif;
				ArrConst(dat%[ich%][:n%], myArray%[:n%]);            'shift data
			endif;
		next;
		View(bh%).BWriteSize(bytes%, dat%[][:pts%]);                'Output it
		stime := stime + pts% * BinSize(list%[ich%]);               'time of next point
	until n% <= 0;                                                 	'until no points left
	end;

'****************************************************************************

	func Quit%()                                                   'Quit button pressed
	View();
	RestoreAll();   
	return 0;
	end;

'****************************************************************************

	proc HideAll()
	var i%;
	
	gFloat%[0] := App(-2);        'number of windows
	for i% := 1 to gFloat%[0] do  'hide all windows and save state
		gFloat%[i%] := View(App(i%)).WindowVisible(0);
	next;
	end

'****************************************************************************

	proc RestoreAll()
	var i%;
	for i% := 1 to gFloat%[0] do           'restore hidden windows
		View(App(i%)).WindowVisible(gFloat%[i%]);
	next;
	end;

'****************************************************************************

